```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the first Weyl algebra $A = A_1(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d,t], polynom=[t]):
```

and the left A-module M finitely presented by the matrix R defined by:

```
> R := evalm([[0,d,0,-1],[d,0,-t,0]]);
```

$$R := \begin{bmatrix} 0 & d & 0 & -1 \\ d & 0 & -t & 0 \end{bmatrix}$$

The rank of the finitely presented left A-module M is:

```
> OreRank(R,A);
```

$$2$$

Since R admits a right inverse S defined by

```
> S := RightInverse(R,A);
```

$$S := \begin{bmatrix} 0 & t \\ 0 & 0 \\ 0 & d \\ -1 & 0 \end{bmatrix}$$

M is a stably free left A-module of rank 2, i.e., a free left A-module of rank 2. Using the fact that the direct sum of $A^{1\times 2}$ and M is isomorphic to $A^{1\times 4}$, and using the Cancellation Theorem, let us compute a basis of M. Let us first compute

```
> X := stackmatrix(R,1-Mult(S,R,A));
```

$$X := \begin{bmatrix} 0 & d & 0 & -1 \\ d & 0 & -t & 0 \\ -d\,t+1 & 0 & t^2 & 0 \\ 0 & 1 & 0 & 0 \\ -d^2 & 0 & 2+d\,t & 0 \\ 0 & d & 0 & 0 \end{bmatrix}$$

which defines the left A-isomorphism g from the direct sum of $A^{1\times 2}$ and M onto $A^{1\times 4}$. Moreover, the direct sum of $A^{1\times 2}$ and M is isomorphic to the left A-module L finitely presented by the matrix P = (0 R) defined by

```
> P := augment(evalm([[0,0],[0,0]]),R);
```

$$P := \begin{bmatrix} 0 & 0 & 0 & d & 0 & -1 \\ 0 & 0 & d & 0 & -t & 0 \end{bmatrix}$$

Similarly, a finite presentation of $A^{1\times 3}$ is given by the matrix R' defined by

```
> Rp := evalm([[0$3]]);
```

$$Rp := \begin{bmatrix} 0 & 0 & 0 \end{bmatrix}$$

and $A^{1\times 4}$ is isomorphic to the left A-module L' finitely presented by the matrix P' = (0 R') defined by

```
>  Pp := augment(evalm([[0]]),Rp);
```

$$Pp := \begin{bmatrix} 0 & 0 & 0 & 0 \end{bmatrix}$$

Let us check again that the left A-homomorphism f from L to L' induced by X is a left A-isomorphism:

```
>  TestIso(P,Pp,X,A);
```

$$true$$

We obtain a left A-isomorphism h_1 from M_1 onto $A^{1\times 3}$ induced by Q1, where M_1 is the left A-module finitely presented by P1, which is isomorphic to the direct sum of A and M, and the matrix P1 = (0 R) is defined by

```
>  P1 := augment(evalm([[0],[0]]),R);
```

$$P1 := \begin{bmatrix} 0 & 0 & d & 0 & -1 \\ 0 & d & 0 & -t & 0 \end{bmatrix}$$

and the matrix Q1 is defined by

```
>  Q1 := Cancellation(Rp,X,A,"splithom");
```

$$Q1 := \begin{bmatrix} d^2 & -t & -d \\ -(d\,t-1)\,d & t^2 & d\,t-1 \\ 1 & 0 & 0 \\ -d^3 & 2+d\,t & d^2 \\ d & 0 & 0 \end{bmatrix}$$

Let us check again that h_1 is a left A-isomorphism:

```
>  TestIso(P1,Rp,Q1,A);
```

$$true$$

Let R" = (0 0), namely,

```
>  Rpp := evalm([[0$2]]);
```

$$Rpp := \begin{bmatrix} 0 & 0 \end{bmatrix}$$

and L" be the left A-module finitely presented by R", which is isomorphic to $A^{1\times 2}$. We obtain a left A-isomorphism h_2 from M onto $A^{1\times 2}$ induced by Q2, where Q2 is the matrix defined by:

```
>  Q2 := Cancellation(Rpp,Q1,A,"splithom");
```

$$Q2 := \begin{bmatrix} t^2+d^2\,t^2+d\,t-1 & d\,t-1+(d\,t-1)\,d^2 \\ -t & -d \\ 2+d\,t+d^3\,t+3\,d^2 & d^2+d^4 \\ -1-d\,t & -d^2 \end{bmatrix}$$

Thus, Q2 is an injective parametrization of M, namely, $ker_A(.\mathrm{Q2}) = A^{1\times 2}$ R,

```
> SyzygyModule(Q2,A);
```

$$\begin{bmatrix} d & 0 & -t & 0 \\ 0 & d & 0 & -1 \end{bmatrix}$$

or equivalently, M is isomorphic to $A^{1\times 4}$ Q2, and Q2 admits a left inverse defined by

```
> T2 := LeftInverse(Q2,A);
```

$$T2 := \begin{bmatrix} 0 & 0 & 1 & 1+d^2 \\ -1 & -t & 0 & -d\,t+1 \end{bmatrix}$$